

# Finding things with Overpass QL

Dave Lambley  
German Perl Workshop 2026


<https://dave.lambley.me.uk/>  
[dave@lambley.me.uk](mailto:dave@lambley.me.uk)

# I watched a climbing-adjacent video



# Where is it?

- Brick railway bridge
- Parallel foot bridge.
- Both over a *navigation*.
- “South Western” livery on train.




# OpenStreetMap Overpass QL

- Query language for OpenStreetMap's data.
- <https://overpass-turbo.eu/>

```
area[name="Guildford"]->.a;  
way(area.a)[waterway][boat!=no];  
out geom;
```

|           |                      |
|-----------|----------------------|
| boat      | yes                  |
| motorboat | yes                  |
| name      | River Wey Navigation |
| waterway  | canal                |






# We've lost canals!

- OpenStreetMap is *tag soup*.
- We use a **union** to find more possible canals.


```
area[name="Guildford"] -> .a;  
(  
  way(area.a)[waterway=canal];  
  way(area.a)[waterway][boat!=no];  
);  
out geom;
```



# We can find bridges over canals

- We can find things near things within a radius.
- **Timeouts! Errors!**
- Overpass QL is imperative not declarative.
- Does everything you ask!
- Limit search area with a “global bbox”.


```
[bbox:51.197, -0.684, 51.261, -0.464];  
area[name="Guildford"]->.a;  
(  
  way(area.a)[waterway=canal];  
  way(area.a)[waterway][boat!=no];  
)->.canals;  
way(around.canals:0)[railway]  
[bridge];  
out geom;
```



# Great, find foot bridges!

- OpenStreetMap *tag soup* also here!


```
(  
  way(around.canals:0)[highway=footway];  
  way(around.canals:0)[highway=path];  
) -> .foot_bridge;
```



# Find foot bridges nearby railway bridges

- Find things nearby foot bridges.
- Compute union with list of railway bridges.

```
way(around.railway_bridge:10) ->.nearby_bridge;  
way.nearby_bridge.foot_bridge;
```



[https://en.wikipedia.org/wiki/Richard\\_Beeching](https://en.wikipedia.org/wiki/Richard_Beeching)  
<https://overpass-turbo.eu/s/2mIU>

```
[bbox:51.197, -0.684, 51.261, -0.464 ];
```

```
area[name="Guildford"]->.a;
```

```
(  
  way(area.a)[waterway][boat][boat!=no];  
  way(area.a)[waterway=canal];  
)->.canals;
```

```
way(around.canals:0)[railway][railway!=abandoned][bridge]  
->.railway_bridge;
```

```
way(around.railway_bridge:10)->.nearby_bridge;
```

```
(  
  way(around.canals:0)[highway=footway];  
  way(around.canals:0)[highway=path];  
)->.foot_bridge;
```

```
way.nearby_bridge.foot_bridge;  
out geom;
```

